

Prolog

Dr. Mattox Beckman

UNIVERSITY OF ILLINOIS AT URBANA-CHAMPAIGN
DEPARTMENT OF COMPUTER SCIENCE

Outline

Introduction

Objectives

Logic

Prolog

Prolog

Queries

Builtin Structures

Activity!

Activity

Objectives

You should be able to...

In this lecture, we will introduce Prolog.

- ▶ Explain how Prolog uses a unification to drive computation.
- ▶ Write some simple programs in Prolog.

Logic

Question: How do you decide truth?

- ▶ Start with some *objects*.
“socrates,” “john,” “mary”
- ▶ Write down some *facts* (true statements) about those objects.
 - ▶ Facts express either properies of the object, or
“socrates is human”
 - ▶ relationship to other objects.
“mary likes john”
- ▶ Write down some *rules* (facts that are true if other facts are true).
“if X is human then X is mortal”
- ▶ Facts and rules can become *predicates*.
“is socrates mortal?”

First-Order Predicate Logic

First-order predicate logic is one system for encoding these kinds of questions.

- ▶ Predicate means that we have functions that take objects and return “true” or “false.”
`human(socrates) .`
- ▶ Logic means that we have *connectives* like and, or, not, and implication.
- ▶ First order means that we have variables (created by “for all” and “there exists”), but that they only work on objects.

$\forall X . \text{human}(X) \rightarrow \text{mortal}(X) .$

History

- ▶ Starting point: First-order predicate logic.
- ▶ Realization: computers can reason with this kind of logic.
- ▶ Impetus was the study of *mechanical theorem proving*
- ▶ Developed in 1970 by Alain Colmerauer and Rober Kowalski and others
- ▶ Uses: databases, expert systems, AI

The Two Questions


What is the nature of data?

Prolog data consists of *facts* about *objects* and logical *rules*.

What is the nature of a program?

A program in Prolog is a set of facts and rules, followed by a *query*.

The Database



```
1 human(socrates).  
2 fatherof(socrates,  
3         jane).  
4 fatherof(zeus,apollo).
```

```
1 connected(c,a).  
2 connected(c,h).  
3 connected(d,b).  
4 connected(d,g).  
5 connected(a,f).  
6 connected(h,f).  
7 connected(b,e).  
8 connected(g,e).
```


Rules

```
1 mortal(X) :- human(X).  
2 human(Y) :- fatherof(X,Y), human(X).  
3  
4 pathfrom(X,Y) :- connected(X,Y).  
5 pathfrom(X,Y) :- connected(X,Z),  
6                     pathfrom(Z,Y).
```

- ▶ Capital letters are variables.
 - ▶ Appearing left of :- means “for all”
 - ▶ Appearing right of :- means “there exists”

$$\forall x. human(x) \rightarrow mortal(x).$$

$$\forall y. (\exists x. fatherof(x,y) \wedge human(x)) \rightarrow human(y)$$

How It Works

Programs are executed by searching the database and attempting to perform unification.

```
1 ?- human(socrates).    -- listed, therefore true
2 ?- mortal(socrates).   -- not listed
```

Relevant rules:

```
1 human(socrates).
2 human(Y) :- fatherof(X,Y), human(X).
3 mortal(X) :- human(X).
```

Socrates is not listed as being mortal, but `mortal(socrates)` unifies with `mortal(X)` if we replace `X` with `socrates`. This gives us a *subgoal*. Replace `X` with `socrates` and try it....

How It Works, Next Step

Replace `X` with `socrates` in this rule:

```
1 mortal(X) :- human(X).
```

to get

```
1 mortal(socrates) :- human(socrates).
```

Since `human(socrates)` is in the database, we know that `mortal(socrates)` is also true.

Another Example

► `?- mortal(jane).`

Another Example

```
► ?- mortal(jane).  
    mortal(X) :- human(X).
```

Another Example

► `?- mortal(jane).`

`mortal(jane) :- human(jane).`

Another Example

▶ `?- mortal(jane).`

`mortal(jane) :- human(jane).`

▶ `human(jane)`

Another Example

► `?- mortal(jane).`

`mortal(jane) :- human(jane).`

► `human(jane)`

`human(Y) :- fatherof(X,Y), human(X).`

Another Example

► `?- mortal(jane).`

`mortal(jane) :- human(jane).`

► `human(jane)`

`human(jane) :- fatherof(X,jane), human(X).`

Another Example

► `?- mortal(jane).`

`mortal(jane) :- human(jane).`

► `human(jane)`

`human(jane) :- fatherof(X,jane), human(X).`

► `fatherof(X,jane)`

Another Example

► `?- mortal(jane).`

`mortal(jane) :- human(jane).`

► `human(jane)`

`human(jane) :- fatherof(X,jane), human(X).`

► `fatherof(X,jane)`

► `fatherof(socrates,jane)`

Another Example

▶ `?- mortal(jane).`

`mortal(jane) :- human(jane).`

▶ `human(jane)`

`human(jane) :- fatherof(socrates,jane), human(socrates).`

▶ `fatherof(X,jane)`

▶ `fatherof(socrates,jane)`

Another Example

► `?- mortal(jane).`

`mortal(jane) :- human(jane).`

► `human(jane)`

`human(jane) :- fatherof(socrates,jane), human(socrates).`

► `fatherof(X,jane)`

► `fatherof(socrates,jane)`

► `human(socrates)`

You Try ...

- ▶ Given the connected rules, try to come up with a predicate `exactlybetween(A,B,C)` that is true when B is connected to both A and C.
- ▶ Now make a predicate `between(A,B,C)` that is true if there's a path from A to B to C.

```
1 exactlybetween(A,B,C) :- connected(A,B), connected(B,C).
```

```
2
```

```
3 between(A,B,C) :- pathfrom(A,B), pathfrom(B,C).
```

More Than Just Yes or No

- Prolog can also give you a list of elements that make a predicate true. Remember unification.

```
1 ?- fatherof(Who,apollo).  
2 Who = zeus ;  
3  
4 ?- pathfrom(c,X).  
5 X = a ;  
6 X = h ;  
7 X = f ;  
8 X = f ;  
9 No
```

The semicolon is entered by the user — it means to keep searching.

Tracing pathfrom

```
1 ?- pathfrom(c,X).  
2 ---> pathfrom(c,Y) :- connected(c,Y).  
3 X = a ;
```

When we hit semicolon, we tell it to keep searching. So we *backtrack* through our database to try again.

```
1 pathfrom(c,Y) :- connected(c,Y).  
2 ---> X = h ;
```

We tell it to try again with this one, too. At this point, we no longer have any rules that say that *c* is connected to something.

Tracing pathfrom, II

```
1 pathfrom(c,Y) :- connected(c,Z), pathfrom(Z,Y).
```

We will first find something in the database that says that *c* is connected to some *Z*, and then check if there is a path between *Z* and *Y*.

We find *a* and *h* as last time. When we check *a*, we check for `pathfrom(a,Y)`, and find that `connected(a,f)` is in the database. The same thing happens for *h*, which is why *f* is reported as an answer twice.

Arithmetic via the `is` Keyword.

```
1 fact(0,1).  
2 fact(N,X) :- M is N-1, fact(M,Y), X is Y * N.  
3 ?- fact(5,X).
```

- ▶ Unify `fact(5,X)` with `fact(N,X)`.

```
fact(5,X) :- M is 5-1, fact(M,Y), X is Y * 5.
```

- ▶ Next compute M.

```
fact(5,X) :- 4 is 5-1, fact(4,Y), X is Y * 5.
```

- ▶ Recursive call sets Y to 24.

```
fact(5,X) :- 4 is 5-1, fact(4,24), X is 24 * 5.
```

- ▶ Compute X.

```
fact(5,120) :- 4 is 5-1, fact(4,24), 120 is 24 * 5.
```

Lists

- ▶ Empty list: `[]`
- ▶ Singleton list: `[x]`
- ▶ List with multiple elements: `[x,y,[a,b],c]`
- ▶ Head and tail representation: `[H|T]`

Differences:

- ▶ Prolog lists are *not* monotonic!

List Example: mylength

The `length` predicate is built in.

```
1 mylength([],0).  
2 mylength([H|T],X) :- mylength(T,Y),  
3                       X is Y + 1.  
4  
5 ?- mylength([2,3,4,5],X).  
6 X = 4 ;  
7 No
```

This example looks like `badfact`, in that the `is` clause happens after the recursion. Why is this safe?

List Example: Sum List

```
1 sumlist([],0).  
2 sumlist([H|T],X) :- sumlist(T,Y),  
3                       X is Y + H.  
4  
5 ?- sumlist([2,3,4,5],X).  
6 X = 14
```

Try writing list product now!

List Example: Append

```
1 myappend([],X,X).  
2 myappend([H|T],X,[H|Z]) :- myappend(T,X,Z).  
3 ?- myappend([2,3,4],[5,6,7],X).  
4 X = [2, 3, 4, 5, 6, 7] ;  
5 No  
6 ?- myappend(X,[2,3],[1,2,3,4]).  
7 No  
8 ?- myappend(X,[2,3],[1,2,3]).  
9 X = [1] ;  
10 No
```

List Example: Reverse

Accumulator recursion works in Prolog, too!

```
1 myreverse(X,Y) :- aux(X,Y, []).  
2 aux([],Y,Y).  
3 aux([HX|TX],Y,Z) :- aux(TX,Y,[HX|Z]).  
4 ?- myreverse([2,3,4],Y).  
5 Y = [4, 3, 2]
```

```
myreverse([2,3,4],Y) → aux([2,3,4],Y,[]) → aux([3,4],Y,[2]) →  
aux([4],Y,[3,2]) → aux([],Y,[4,3,2]) → aux([],Y,[4,3,2]) →  
myreverse([2,3,4],[4,3,2])
```


Activity

- ▶ Write the Fibonacci predicate. Let $F_0 = 0$ and $F_1 = 1$.
- ▶ Make sure you can write it the exponential way.
- ▶ Can you write it the linear way?

Solution

- ▶ Fibonacci predicate: exponential complexity:

```
1 fib(0,0).  
2 fib(1,1).  
3 fib(N,X) :- N1 is N - 1,  
4             fib(N1,X1), N2 is N - 2,  
5             fib(N2,X2), X is X1 + X2.
```

- ▶ Fibonacci predicate: linear complexity:

```
1 lfibx(0,F1,F2,A) :- A is F2.  
2 lfibx(N,F1,F2,A) :- N1 is N - 1,  
3                     F3 is F1 + F2,  
4                     lfibx(N1,F2,F3,A).  
5 lfib(N,A) :- lfibx(N,1,0,A).
```