Monads

Dr. Mattox Beckman

University of Illinois at Urbana-Champaign
Department of Computer Science

Objectives

- Describe the problem that monads attempt to solve.
- Know the three monad laws.
- Know the syntax for declaring monadic operations.
- Be able to give examples using the Maybe and List monads.

Introducing Monads

- Monads are a way of defining computation.
- A *monad* is a container type *m* along with two functions:
  - `return :: a -> m a`
  - `bind :: m a -> (a -> m b) -> m b`
  - In HASKELL, `bind` is written as `>>=`
- These functions must obey three laws:

  Left identity `return a >>= f` is the same as `f a`.

  Right identity `m >>= return` is the same as `m`.

  Associativity `(m >>= f) >>= g` is the same as `m >>= (\x -> f x >>= g)`.

Understanding Return

- `return :: a -> m a`
- The `return` keyword takes an element and puts it into a monad.
- This is a one-way trip!
- Very much like `pure` in the applicative type class.

```haskell
instance Monad Maybe where
  return a = Just a
instance Monad [] where
  return a = [a]
instance Monad (Either a) where
  return a = Right a
```

Understanding Bind

- All the magic happens in bind.
- `bind :: m a -> (a -> m b) -> m b`
  - The first argument is a monad.
  - The second argument takes a monad, unpacks it, and repackages it with the help of the function argument.
    - Exactly *how* it does that is the magic part.

Bind for Maybe

```
Nothing  >>= f  = Nothing
(Just a) >>= f = f a
   -- Remember that f returns a monad
```

Motivation

- They are similar to continuations, but even more powerful.
- They are also related to the applicative functors from last time.
- Consider this program:

```
inc1 a = a + 1
r1 = inc1 <$> Just 10 -- result: Just 11
r2 = inc1 <$> Nothing -- result: Nothing
```

But what if we have functions like this?

```
inc2 a = Just (a+1)
recip a | a =/ 0     = Just (1/a)
        | otherwise = Nothing
```

How can we pass a Nothing to it? How can we use what we get from it?

Notice the Pattern

- Applicatives take the values out of the parameters, run them through a function, and then repackage the result for us.
- The functions have no control: the applicative makes all the decisions.
- Monads let the functions themselves decide what should happen.

A Calculator, with Monads

```
minc x = x >>= (\xx -> return (xx + 1))
madd a b = a >>= (\aa ->
               b >>= (\bb -> return (aa+bb)))
-- but wait!!!
```

- Okay, the above code works, but here's a better way.
- First define functions `lift` to convert a function to monadic form for us!

These are part of Control.Monad:

```
liftM f a = a >>= (\aa -> return (f aa))
liftM2 f a b = a >>= (\aa ->
                  b >>= (\bb -> return (f aa bb)))
```

Continued

Lifting

```haskell
minc = liftM inc
madd = liftM2 add
msub = liftM2 sub
mdiv a b = a >>= (\aa ->
              b >>= (\bb ->
                if bb == 0 then fail "/0"
                           else return (aa `div` bb)))
```

- `fail` is another useful monadic function, defined in the `MonadFail` typeclass.
- Here it's defined as `Nothing`.

The Maybe Monad

- Here is the complete monad definition for Maybe.

Maybe Monad

```
instance Monad Maybe where
  return = Just

  (>>=) Nothing f  = Nothing
  (>>=) (Just a) f = f a

  fail s = Nothing
```

Example with Maybe

```
Prelude> minc (Just 10)
Just 11
Prelude> madd (minc (Just 10)) (Just 20)
Just 31
Prelude> mdiv (minc (Just 10)) (minc (Just 2))
Just 3
Prelude> minc (mdiv (minc (Just 10)) (minc (Just 2)))
Just 4
Prelude> minc (mdiv (minc (Just 10)) (Just 0))
Nothing
```

The List Monad

- Lists can be monads too. The trick is deciding what bind should do.

List Monad

```
instance Monad [] where
  return a = [a]

  (>>=) [] f   = []
  (>>=) xs f   = concatMap f xs

  fail s = []
```

- Note that we do not have to change *anything* in our lifted calculator example!

Example with List

```
Prelude> minc [1,2,3]
[2,3,4]
Prelude> madd [1,2,3] [10,200]
[11,201,12,202,13,203]
Prelude> minc (mdiv [10] [0])
[]
Prelude> minc (mdiv [10] [0,2,5])
[5,2]
```